Computer input and output to accompany Statistical Methods for the Social and Behavioural Sciences: A Model-Based Approach by David B. Flora
[bookmark: _GoBack]This document © 2017 by David B. Flora and Joo Ann Lee
SAS syntax and output for Chapter 5: Using Regression to Model Mediation
Much of the material in this document builds on the SAS syntax and output from earlier chapters. If necessary, refer back to those documents for reminders.

Begin by importing the ‘DunnBiesanz.dat’ file (for the self-presentation example) into SAS:
	data ch5data;
 infile 'c:\DunnBiesanz.dat' firstobs=2; 
 input id $ Cond SP Affect;
 drop id;
run;



Histograms of self-presentation (variable name 'SP') and positive affect (variable name 'Affect'). These histograms form Figure 5.2:

	title Histograms of Affect;
proc sgplot data=ch5data;
 histogram Affect; 
run;

title Histograms of SP;
proc sgplot data=ch5data;
 histogram SP; 
run;

Side-by-side boxplot of self-presentation by manipulation (variable name 'Cond') and scatterplot of positive affect by self-presentation. These plots form Figure 5.3:

	proc sgplot data=ch5data;
 vbox SP / group=Cond;
 yaxis label="Self-presentation scores";
 keylegend / title="Manipulation";
run;

proc sgplot data=ch5data;
 scatter x=SP y=Affect; 
run;

Regression models for simple-indirect effect model
First, regress the mediator ('SP') on the main predictor ('Cond'):
	proc reg data=ch5data plots=none;
 model SP = Cond / clb;
run;
quit;




In the output above, the estimate for the 'Cond' predictor corresponds to the alpha-hat estimate; results match those for the self-presentation outcome in Table 5.1.
Next, regress the main dependent variable ('Affect') on the mediator ('SP') and main predictor ('Cond'):
	proc reg data=ch5data plots=none;
 model Affect = SP Cond / clb;
run;
quit;




In the output above, the estimate for the 'SP' predictor corresponds to the beta-hat estimate; results match values for the positive affect outcome in Table 5.1.

Percentile bootstrap confidence interval for the indirect effect
The 'process' macro from http://www.processmacro.org contains functions for mediation/indirect effect analyses. It is necessary to first run the full macro code in SAS before using the ‘process’ macro.
The first argument of the 'process' macro is the name of the dataset. 
The second argument is the name of variables in the model, followed by the variables’ specific role in the mediation model.
‘total = 1’ shows the output for the total effect while ‘normal=1’ produces output for the statistical test for indirect effect.
The 'boot' option is used to indicate the number of bootstrap samples. ‘model=4’ specifies a simple mediation model:
	%process (data=ch5data,vars=SP Cond Affect, y=Affect, x=Cond, m=SP, total=1, normal=1, boot=10000, model=4);



The first sections of the process output are redundant with the regression models presented earlier; skip ahead to the “TOTAL, DIRECT AND INDIRECT EFFECTS”:
	





	


	




The 'Indirect effect of X on Y' estimate should correspond to the product-of-coefficients indirect effect estimate given in Chapter 5.
The corresponding 'BootLLCI' and 'BootULCI' values indicate the limits of the 95% percentile bootstrap confidence interval for the indirect effect; these values don’t exactly match the bootstrap confidence interval reported in Chapter 5 because bootstrapping necessarily involves a degree of sampling error.

Moving on to the research example for modeling a non-experimental indirect effect (heavy alcohol use example):
Import the drink.txt data into SAS:
	data ch5data2;
 infile 'c:\drink.txt' firstobs=2 dlm=' ';
 input id coa alcuse ext numalc;
 drop id;
run;


Note that this is the same dataset used in Chapter 4.
Plots of bivariate associations between externalizing and parental alcoholism and between alcohol use and externalizing. These plots form Figure 5.5:

	proc sgplot data=ch5data2;
 vbox ext / group=coa;
 yaxis label="Externalizing";
 keylegend / title="Parental alchoholism (1=COA)";
run;

proc sgplot data=ch5data2;
 scatter x=ext y=alcuse; 
 yaxis label="Alcohol use";
 xaxis label="Externalizing";
run;

Regression models for simple-indirect effect model.
First, regress the mediator ('ext') on the main predictor ('coa'):
	proc reg data=ch5data2 plots=none;
 model ext = coa / clb;
run;
quit;




In the output above, the estimate for the 'coa' predictor corresponds to the alpha-hat estimate; results match those for the Externalizing outcome in Table 5.2.
Next, regress the main dependent variable ('alcuse') on the mediator ('ext') and main predictor ('coa'):
	proc reg data=ch5data2 plots=none;
 model alcuse = ext coa / clb;
run;
quit;




In the output above, the estimate for the 'ext' predictor corresponds to the beta-hat estimate; results match values for Heavy alcohol use outcome in Table 5.2.
Next, use the 'process' macro to get a bootstrap CI for the indirect effect:
	%process (data=ch5data2,vars=alcuse ext coa, y=alcuse, x=coa, m=ext, total=1, normal=1, boot=10000, model=4);


	




	


	




Again, the 'Indirect effect of X on Y' estimate corresponds to the product-of-coefficients indirect effect estimate given in Chapter 5.
The corresponding 95% bootstrap CI should be approximately (0.02, 0.20), but may differ slightly because of the random sampling error inherent to the bootstrapping procedure.

Moderated indirect effect
Regress the main outcome variable on mediator ('ext'), main predictor ('coa'), and their product.
Note that this model is identical to the dichotomous by continuous interaction example from Chapter 4:
	data ch5data2;
set ch5data2;
 extcoa = ext*coa;
run;

proc reg data=ch5data2 plots=none;
 model alcuse = ext coa extcoa / clb;
run;
quit;





In the output above, the estimate for the 'ext' predictor corresponds to the beta-hat estimate (i.e., simple slope) for control participants.
Next, use the 'process' macro to get bootstrap CI for the separate indirect effects for the control and COA groups:
	%process (data=ch5data2,vars=alcuse ext coa, y=alcuse, x=coa, m=ext, total=1, normal=1, boot=10000, model=74);

	



In the output above, because of the interaction, the indirect effect estimate of 0.02, along with the associated 95% percentile bootstrap CI, pertains only to control participants (extending the idea of the simple slope analysis from Chapter 4 to the current moderated mediation context).
To obtain the indirect effect for the COA group, we first need to re-code the ‘coa’ variable, as was done in Chapter 4 to probe the externalizing by COA interaction:

	data ch5data2;
set ch5data2;
 ncoa=abs(coa-1);
 extncoa=ext*ncoa;
run;



Then we can re-run the PROCESS command using this re-coded variable:

	%process (data=ch5data2,vars=alcuse ext ncoa, y=alcuse, x=ncoa, m=ext, total=1, normal=1, boot=10000, model=74);


Now we obtain the separate indirect effect estimate for COA participants, along with the bootstrap confidence interval:

